//=======================================================================
// The Primary Siege Mutator
//=======================================================================
class TFMainMut extends UTMutator
HideDropDown;

var int ID;

//carry over properties to next mutator
function bool AlwaysKeep(Actor Other)
{
	if ( NextMutator != None )
		return ( NextMutator.AlwaysKeep(Other) );
	return false;
}

//replace actors
function bool CheckReplacement(Actor Other)
{
    local TFStation b;
    local vector v;

    //replace all super pickups/weapons with minerals
	if (Other.IsA('UTPickupFactory_UDamage') || Other.IsA('UTPickupFactory_SuperHealth') || Other.IsA('UTPickupFactory_SlowField') || Other.IsA('UTPickupFactory_Invulnerability') || Other.IsA('UTPickupFactory_JumpBoots') || Other.IsA('UTPickupFactory_Invisibility') || Other.IsA('UTPickupFactory_Berzerk') || (UTWeaponPickupFactory(other)!=none && (UTWeaponPickupFactory(other).WeaponPickupClass==class'UTGame.UTWeap_Redeemer')))
	{
        v=other.Location;
	    if(UTPickupFactory(other)!=none) v.Z+=UTPickupFactory(other).BobBaseOffset*1.25;
		spawn(class'TaskForces.TFMinerals',,,v);
	}

    //destroy all pickups/pickup bases
	if (Other.IsA('PickupFactory'))
	{
		Other.SetHidden(true);
        return false;
	}

    //replace CTF flag base with base core and set the cores team
    if (Other.IsA('UTCTFBase'))
    {
	    b=spawn(class'TaskForces.TFBaseCore_Content',,,other.location);
        b.setinitialteam(UTCTFBase(other).DefenderTeamIndex);

        Other.SetHidden(true);
        return false;
    }

	return true;
}

defaultproperties
{

}
